LIGO-T890015-00-R

Round Trip Loss for a Sapphire Crystal in a Loss-Less Cavity

Alex Abramovici

26 July 1989

Abstract

A sapphire crystal with end faces at the Brewster angle is assumed inside a cavity with loss-less mirrors. The round trip loss due to the birefringence of the crystal is estimated by using the Jones polarization matrix method. The resulting loss contour plots show how the crystal has to be cut in order to keep the losses below a specified level.

1 Introduction

As an example of a cavity less affected by heating effects due to beam absorption in the mirror coatings, R. W. P. Drever has suggested a geometry where the mirrors have nontransparent substrates with good heat conductivity, while the coupling to and from the cavity is ensured by a transparent plate, placed between the mirrors, approximately at Brewster's angle.

The coupling plate should not introduce losses much in excess of the mirror losses. Therefore, the material the plate is made of should have a low absorption coefficient. Other requirements are good heat conductivity, the ability to sustain a superpolish, etc. One candidate material, fused silica, has low loss, can be superpolished, but has a relatively low thermal conductivity. It has therefore been suggested that the coupling plate be made of sapphire, which has higher thermal conductivity. Unlike fused silica, which is a glass that can be made with very low residual birefringence, sapphire is a birefringent crystal. As a consequence, except for special orientations of the optical axis[1] a polarized light beam incident on a sapphire plate at Brewster's angle incures a certain level of loss due to the polarization state of the light being modified by the plate. The resulting round trip loss for a crystal in a loss-less cavity is estimated in what follows by using the Jones polarization matrix formalism[2]. The resulting loss contour plots allow to define the accuracy required in cutting the crystal, in order for the birefringence induced loss to be below a specified level.

2 Jones Matrix for a Birefringent Plate at the Brewster Angle

Assume that the birefringent plate is at a Brewster angle with respect to the incident beam. Since the sapphire birefringence is small ($\delta n = 0.0082$), it will be assumed that the Brewster angle is the same, independent of beam polarization. Denote α the angle between the optical axis and the direction of propagation of the beam in the crystal. Furthermore, denote

[1] e. g. when the optical axis is in the plane of incidence

[2] R. C. Jones, *A New Calculus for the Treatment of Optical Systems*, J. Opt. Soc. of America,**31**,488 (1941)

β the angle between the plane of incidence and the plane defined by the optical axis and the direction of propagation of the light in the crystal.[3] It can be shown that if α is small, then, for a uniaxial crystal the effective birefringence seen by the propagating light can be written as:

$$\delta n_r = \alpha^2 \delta n \tag{1}$$

The cofiguration considered here can be understood in terms of three different Jones matrices. At the Brewster face, one polarization passes unchanged, while the other one is attenuated due to partial reflection. The corresponding matrix is:

$$\mathcal{B}(n) = \begin{pmatrix} 1 & 0 \\ 0 & \frac{2n}{1+n^2} \end{pmatrix} \tag{2}$$

The fact that the plane of incidence is rotated with respect to the principal axes of the crystal is described by the rotation matrix:

$$\mathcal{R}(\beta) = \begin{pmatrix} \cos\beta & \sin\beta \\ -\sin\beta & \cos\beta \end{pmatrix} \tag{3}$$

Finally, the wave plate effect of the crystal is described by:

$$\mathcal{D}(\delta n_r) = \begin{pmatrix} 1 & 0 \\ 0 & e^{i\phi} \end{pmatrix} \tag{4}$$

with $\phi = 2\pi l \delta n_r / \lambda$, where l is the geometrical length of the beam path through the crystal and λ is the wave length of the light.

The complete Jones matrix for a round trip through the cavity containing the crystal is:

$$\mathcal{M} = \mathcal{B}\mathcal{R}(\beta)\mathcal{D}\mathcal{R}(-\beta)\mathcal{B}\mathcal{B}\mathcal{R}(\beta)\mathcal{D}\mathcal{R}(-\beta)\mathcal{B} \tag{5}$$

3 The Round Trip Losses

When a light beam with a certain polarization carries out a round trip through the cavity containing the birefringent element, its polarization state

[3] β will henceforth be called **azimuth**

is generally changed. However, there are two polarization states which are left unchanged after a round trip, except, possibly, for undergoing attenuation. These states are the eigenvectors χ_i of the Jones matrix $\mathcal{M}$, and the corresponding amplitude transmission coefficients are the eigenvalues η_i:

$$\mathcal{M}\chi_i = \eta_i \chi_i \tag{6}$$

The loss for a round trip then is:

$$L_i = | \eta_i |^2 \tag{7}$$

In a ring-down time measurement of the losses, the polarization state with higher loss will decay faster, so that the actually measured loss will correspond to the low loss state. The losses for the low loss state, for a beam path of 2.54 cm through a sapphire crystal with Brewster faces, are shown in the accompanying figure, as a function of the angles α and β. The contour plots can be used to derive the accuracy requirements on cutting the crystal, corresponding to a given upper limits of the losses due to birefringence.



Loss contours
Length: 2.54 cm
50 ppm
100 ppm
1000 ppm
10,000 ppm
100,000 ppm
Azimuth angle of e/o axis (degrees)
Optical/geometrical axes angle (degrees)
0
10
20
30
40
0
.1
.2
.3
.4
.5

```
      program losc
c
c
c
c Program calculates the rpund trip loss of a brewster
c plate made of a uniaxial crystal with the axis at a
c small angle with respect to the light beam.
c
c Output is a contour plot.
c
c 6 July 1989
c
c Alex
c
c
      dimension beta(31),x(31),y(31),alfa(31),t(31,31)
      complex psi(31),a(31,31),b(31,31),c(31,31),d(31,31),r1(31,31),
     *r2(31,31),u1,u2
      character*70 co(40)
      pi=2*acos(0.)
      ambda=514.5e-7
c
c Input data:
c
      print 1
    1 format(1x,"Average index of refraction")
      read*,enn
      print 2
    2 format(1x,"birefringence")
      read*,deltan
      print 3
    3 format(1x,"Range of azimutal angles of optical axis (degrees)")
c
c azimutal angle is in plane perpendicular to light beam
c
      read*,bmin,bmax
      print 4
    4 format(1x,"Range of angles between optical axis and beam")
      read*,almin,almax
      print 5
    5 format(1x,"Geometrical beam path through crystal (cm)")
      read*,el
c
c open output file
c
      open(8,file="losc.out")
      rewind(8)
c
      en=2.*enn/(1.+enn*enn)
c
c Calculate azimutal angles, build arrays of sines and cosines
c
      step=(bmax-bmin)/30.*pi/180.
      do 10 i=1,31
      beta(i)=step*(i-1)+bmin*pi/180.
      x(i)=cos(beta(i))
   10 y(i)=sin(beta(i))
c
```

```
c Build array of axis angles
c
     step=(almax-almin)/30.*pi/180.
     do 20 i=1,31
     alfa(i)=step*(i-1)+almin*pi/180.
  20 continue
c
c Build array of phase factors
c
     do 30 i=1,31
     p=2.*pi/ambda*el*alfa(i)**2*deltan
     psi(i)=cmplx(0,p)
  30 psi(i)=exp(psi(i))
c
c Calculate the elements of the Jones matrix
c
     do 50 j=1,31
     do 40 i=1,31
     a(i,j)=(x(j)**2+psi(i)*y(j)**2)**2
    *+en**2*x(j)**2*y(j)**2*(1-psi(i))**2
     b(i,j)=+en*x(j)*y(j)*(1-psi(i))*(x(j)**2+psi(i)*y(j)**2)
    *+en**3*x(j)*y(j)*(1-psi(i))*(y(j)**2+psi(i)*x(j)**2)
     c(i,j)=b(i,j)
     d(i,j)=en**2*x(j)**2*y(j)**2*(1-psi(i))**2
    *+en**4*(y(j)**2+psi(i)*x(j)**2)**2
  40 continue
  50 continue
c
c calculate the array of eigenvalues
c
     do 70 j=1,31
     do 60 i=1,31
     u1=a(i,j)+d(i,j)
     u2=u1**2-4.*(a(i,j)*d(i,j)-b(i,j)*c(i,j))
     u2=u2**(.5)
     r1(i,j)=.5*(u1+u2)
     r2(i,j)=.5*(u1-u2)
  60 continue
  70 continue
c
c Calculate the array of highest transmissions
c
     do 90 j=1,31
     do 80 i=1,31
     f=abs(r1(i,j))
     g=abs(r2(i,j))
     g=max(f,g)
     t(i,j)=g*g
  80 continue
  90 continue
c
c Print input data
c
     write(8,150) enn
 150 format(1x,"Average index of refraction:",f5.3/)
     write(8,151) deltan
 151 format(1x,"Birefringence:",f6.4/)
     write(8,152) bmin,bmax
```

```
 152 format(1x,"Range of azimutal angles:",f5.1," to ",f5.1,
   *" degrees"/)
    write(8,153) almin,almax
 153 format(1x,"Range of axis angles:    ",f5.1," to ",f5.1,
   *" degrees"/)
    write(8,154) el
 154 format(1x,"Geometrical beampath:    ",f6.2," cm"///)
c
c  Print the array of highest transmissions, with axis angles running
c  horizontally (left-right) and azimuths runing vertically (top-bottom)
c
    write(8,155) ((t(i,j),i=1,31,6),j=1,31,6)
 155 format(1x,6f13.8)
    close(8)
    co(1)='terminal 6'
    co(2)='limits 0 .5 0 45'
    co(3)='lweight 3'
    co(4)='box'
    co(5)='xlabel Optical/geometrical axes angle (degrees)'
    co(6)='ylabel Azimuth angle of e/o axis (degrees)'
    co(7)='contour 5 .99995 .9999 .999 .99 .9'
    co(8)='connect'
    co(9)='relocate .33 30'
    co(10)='putlabel 3 Loss contours'
    co(11)='angle -90'
    co(12)='relocate .063 43'
    co(13)='putlabel 3 50 ppm'
    co(14)='relocate .1 43'
    co(15)='putlabel 3 100 ppm'
    co(16)='relocate .142 43'
    co(17)='putlabel 3 1000 ppm'
    co(18)='relocate .26 43'
    co(19)='putlabel 3 10,000 ppm'
    co(20)='relocate .46 43'
    co(21)='angle -70'
    co(22)='putlabel 3 100,000 ppm'
    co(23)='angle 0'
    co(24)='relocate .33 28'
    co(25)='putlabel 3 Length: 2.54 cm'
    call mongo(25,co,31,31,t)
    stop
    end
```